program:




```
program  ::= ( statement | NEWLINE )* EOF
```

no references

statement:




```
statement
         ::= declaration_stmnt
           | if_stmnt
           | while_stmnt
           | do_stmnt
           | for_stmnt
           | print_stmnt
           | return_stmnt
           | input_stmnt
           | import_stmt
           | pass_stmnt
           | break_stmnt
           | continue_stmnt
           | expression_stmnt
```

referenced by:

- block
- program

declaration_stmnt:




```
declaration_stmnt
         ::= function_declaration
           | variable_declaration
```

referenced by:

- statement

function_declaration:




```
function_declaration
        ::= 'def' identifier '(' ( identifier ( ',' identifier )* )? ')' block
```

referenced by:

- declaration_stmnt

variable_declaration:




```
variable_declaration
        ::= var_type identifier ( '=' assignment_expr )? ( ',' identifier ( '=' assignment_expr )? )* NEWLINE
```

referenced by:

- declaration_stmnt

var_type:




```
var_type ::= 'char'
           | 'int'
           | 'float'
           | 'str'
           | 'list'
```

referenced by:

- variable_declaration

if_stmnt:




```
if_stmnt ::= 'if' expression block ( 'else' block )?
```

referenced by:

- statement

while_stmnt:


```
while_stmnt
        ::= 'while' expression block
```

referenced by:

- statement

do_stmnt:


```
do_stmnt ::= 'do' block 'while' expression NEWLINE
```

referenced by:

- statement

for_stmnt:


```
for_stmnt
        ::= 'for' IDENTIFIER 'in' sequence block
```

referenced by:

- statement

print_stmnt:




```
print_stmnt
        ::= 'print' '-raw'? ( assignment_expr ( ',' assignment_expr )* )? NEWLINE
```

referenced by:

- statement

return_stmnt:




```
return_stmnt
        ::= 'return' expression? NEWLINE
```

referenced by:

- statement

input_stmnt:




```
input_stmnt
        ::= 'input' string? identifier ( ',' string? identifier )* NEWLINE
```

referenced by:

- statement

import_stmt:




```
import_stmt
        ::= 'import' string_literal NEWLINE
```

referenced by:

- statement

pass_stmnt:




```
pass_stmnt
        ::= 'pass' NEWLINE
```

referenced by:

- statement

break_stmnt:




```
break_stmnt
        ::= 'break' NEWLINE
```

referenced by:

- statement

continue_stmnt:


```
continue_stmnt
        ::= 'continue' NEWLINE
```

referenced by:

- statement

block:




```
block    ::= NEWLINE INDENT statement+ DEDENT
```

referenced by:

- do_stmnt
- for_stmnt
- function_declaration
- if_stmnt
- while_stmnt

expression_stmnt:




```
expression_stmnt
        ::= expression NEWLINE
```

referenced by:

- statement

expression:




```
expression
        ::= comma_expr
          | assignment_expr
```

referenced by:

- do_stmnt
- expression_stmnt

- if_stmnt
- primary_expr
- return_stmnt
- while_stmnt

comma_expr:




```
comma_expr
        ::= assignment_expr ( ',' assignment_expr )+
```

referenced by:

- expression

assignment_expr:




```
assignment_expr
        ::= logical_or_expr ( ( '=' | '+=' | '-=' | '*=' | '\=' | '%=' ) assignment_expr )*
```

referenced by:

- assignment_expr
- comma_expr
- expression
- function_call
- list_literal
- print_stmnt
- variable_declaration

logical_or_expr:




```
logical_or_expr
        ::= logical_and_expr ( 'or' logical_or_expr )*
```

referenced by:

- assignment_expr
- index
- logical_or_expr
- method
- slice

logical_and_expr:




```
logical_and_expr
        ::= equality_expr ( 'and' logical_and_expr )*
```

referenced by:

- logical_and_expr
- logical_or_expr

equality_expr:




```
equality_expr
        ::= relational_expr ( ( '==' | '!=' | '<>' | 'in' ) equality_expr )*
```

referenced by:

- equality_expr
- logical_and_expr

relational_expr:




```
relational_expr
        ::= addition_expr ( ( '<' | '>' | '<=' | '>=' ) relational_expr )*
```

referenced by:

- equality_expr
- relational_expr

addition_expr:




```
addition_expr
        ::= multiplication_expr ( ( '+' | '-' ) addition_expr )*
```

referenced by:

- addition_expr
- relational_expr

multiplication_expr:




```
multiplication_expr
        ::= unary_expr ( ( '*' | '/' | '%' ) multiplication_expr )*
```

referenced by:

- addition_expr
- multiplication_expr

unary_expr:




```
unary_expr
        ::= ( '+' | '-' | '!' )? primary_expr
```

referenced by:

- multiplication_expr

primary_expr:




```
primary_expr
        ::= ( function_call | variable | literal | '(' expression ')' ) subscript? ( '.' method )?
```

referenced by:

- unary_expr

function_call:




```
function_call
        ::= identifier '(' ( assignment_expr ( ',' assignment_expr )* )? ')'
```

referenced by:

- primary_expr

subscript:




```
subscript
        ::= '[' ( index | slice ) ']'
```

referenced by:

- primary_expr

index:




```
index    ::= logical_or_expr
```

referenced by:

- subscript

slice:




```
slice    ::= logical_or_expr? ':' logical_or_expr?
```

referenced by:

- subscript

method:




```
method   ::= identifier '(' ( logical_or_expr ( ',' logical_or_expr )* )? ')'
```

referenced by:

- primary_expr

identifier:




```
identifier
         ::= alphabetic ( alphabetic | digit | '_' )*
```

referenced by:

- function_call
- function_declaration
- input_stmnt
- method
- variable
- variable_declaration

variable:




```
variable ::= identifier
```

referenced by:

- primary_expr

literal:




```
literal  ::= numeric_literal
           | character_literal
           | string_literal
           | list_literal
```

referenced by:

- primary_expr

numeric_literal:




```
numeric_literal
         ::= number
```

referenced by:

- literal

character_literal:




```
character_literal
         ::= "'" character "'"
```

referenced by:

- literal

string_literal:




```
string_literal
         ::= string
```

referenced by:

- import_stmt
- literal

list_literal:




```
list_literal
         ::= '[' ( assignment_expr ( ',' assignment_expr )* )? ']'
```

referenced by:

- literal

string:




```
string   ::= '"' character* '"'
```

referenced by:

- input_stmnt
- string_literal

character:




```
character
         ::= 'any character except single quote'
           | escape
```

referenced by:

- character_literal
- string

alphabetic:




```
alphabetic
         ::= [a-zA-Z]
```

referenced by:

- identifier

escape:




```
escape   ::= '\' [bnfrtv\'"0]
```

referenced by:

- character

number:




```
number   ::= integer
           | float
```

referenced by:

- numeric_literal

integer:




```
integer  ::= digit+
```

referenced by:

- float
- number

float:




```
float    ::= integer '.' integer? ( ( 'e' | 'E' ) ( '+' | '-' )? integer )?
```

referenced by:

- number

digit:




```
digit    ::= [0-9]
```

referenced by:

- identifier
- integer
